

# PRINT

```

PRINT [(rep)] [NOTITLE] [NOHDR] [(statement-parameters)]
  {
    {
      [nX] { 'text' [(attributes)] }
      [nT] { 'c'(n) [(attributes)] }
      [ / ] ... [ '=' operand1 [(parameters)] ] ...
    }
  }

```

Operand	Possible Structure				Possible Formats										Referencing Permitted	Dynamic Definition	
Operand1	S	A	G	N	A	N	P	I	F	B	D	T	L	G	O	yes	no

Related Statement: WRITE

## Function

The PRINT statement is used to produce output in free format.

The PRINT statement differs from the WRITE statement in the following aspects:

- The output for each operand is written according to the value content rather than the length of the operand. Leading zeros for numeric values and trailing blanks for alphanumeric values are suppressed. The session parameter AD defines whether numeric values are printed left or right justified. With AD=L, the trailing blanks of a numeric value are suppressed. With AD=R, the leading blanks of a numeric value are printed.
- If the resulting output exceeds the current line size (LS parameter), the output is continued on the next line as follows:  
An alphanumeric constant or the content of an alphanumeric variable (without edit mask) is split at the rightmost blank or character which is neither a letter nor a numeric character contained on the current line. The first part of the split value is output to the current line, and the second part is written to the next line. For all other operands, the entire value is written to the next line.

## Report Specification - rep

The notation (*rep*) may be used to specify the identification of the report for which the PRINT statement is applicable. A value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified. If (*rep*) is not specified, the PRINT statement will be applicable to the first report (report 0).

## NOTITLE

Natural generates a single title line for each page resulting from a PRINT statement. This title contains the page number, the time of day, and the date. Time of day is set at the beginning of the session (TP mode) or at the beginning of the job (batch mode). This title line may be overridden by using a WRITE TITLE statement or may be suppressed by specifying the NOTITLE clause in the PRINT statement.

PRINT NAME	(default title will be produced)
PRINT NAME	
WRITE TITLE 'USER TITLE'	(user title will be produced)
PRINT NOTITLE NAME	(no title will be produced)

If the NOTITLE option is used, it applies to all DISPLAY, PRINT and WRITE statements within the same object which write data to the same report.

## NOHDR

The PRINT statement itself does not produce any column headers. However, if you use the PRINT statement in conjunction with a DISPLAY statement, you can use the NOHDR option of the PRINT statement to suppress the column headers generated by the DISPLAY statement: the NOHDR option only takes effect if the PRINT statement is executed **after** a DISPLAY statement, the output spans more than one page, and the execution of the PRINT statement causes a new page to be output. Without the NOHDR option, the column headers of the DISPLAY statement would be output on this new page; with NOHDR they will not.

## statement-parameters

One or more parameters, enclosed within parentheses, may be specified immediately after the PRINT statement or an element being displayed.

Each parameter specified in this manner will override any previous parameter specified in a GLOBALS command, SET GLOBALS or FORMAT statement. If more than one parameter is specified, the parameters must be separated from one another by one or more blanks. A parameter entry must not be split between two statement lines.

The parameter settings applied here will only be regarded for variable fields, but they have no effect on text-constants. If you would like to set field attributes for a text-constant, they have to be set explicitly for this element.

### Example:

```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>          /*      Output
END-DEFINE      /*      Produced
*              /*      -----
PRINT           'Text'                VARI          /*      Text 1234
PRINT (PM=I)   'Text'                VARI          /*      Text 4321
PRINT          'Text' (PM=I)         VARI (PM=I)  /*      txeT 4321
PRINT          'Text' (PM=I)         VARI          /*      txeT 1234
END

```

The individual parameters are described in the section Session Parameters of the Natural Reference documentation.

## Field Positioning, Text, Attribute Assignment

$\left\{ \left[ \begin{array}{l} nX \\ nT \\ / \end{array} \right] \dots \left[ [='] \text{operand1} \left[ \left[ \text{parameters} \right] \right] \right] \dots \right\}$	$\left\{ \text{'text' } \left[ \left[ \text{attributes} \right] \right] \right\}$
	$\left\{ \text{'c'(n) } \left[ \left[ \text{attributes} \right] \right] \right\}$
	$\left\{ \text{'=' } \text{operand1} \left[ \left[ \text{parameters} \right] \right] \right\}$

## Field Positioning Notations

### nX

The *nX* notation is used to insert *n* spaces between columns.

**Note:** (for Mainframes Only)

This notation inserts *n* spaces between columns. *n* must not be "0".

```
PRINT NAME 5X SALARY
```

### nT

The *nT* notation causes positioning (tabulation) to print position *n*. Backward positioning results in a line advance.

```
PRINT 25T NAME 50T SALARY
```

(causes NAME to print beginning in position 25 and SALARY to print beginning in position 50).

/

A slash causes a line advance when placed between fields or text elements.

```
PRINT NAME / SALARY
```

## Text/Attribute Assignment

### 'text'

*text* is displayed as text.

```
PRINT 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT
```

### 'c' (n)

Identical to '*text*', except that the specified character *c* is displayed *n* times.

```
PRINT '*' (5) '=' NAME
```


### '='

If '=' is placed immediately before the field, the field name is output immediately before the field value.

```
PRINT '=' NAME
```

## attributes

The display and color attributes to be used for text/field display:



1. Display attributes (see the session parameter AD in the Natural Reference documentation).
2. Color attributes (see the session parameter CD in the Natural Reference documentation).

## operand1

As *operand1* you specify the field to be printed.

## parameters

One or more parameters, enclosed within parentheses, may be specified immediately after *operand1*. Each parameter specified in this manner will override any previous parameter specified in a GLOBALS command, SET GLOBALS or FORMAT statement. If more than one parameter is specified, one or more blanks must be placed between each entry. An entry must not be split between two statement lines.

## Example

```

/* EXAMPLE 'PRTEX1': PRINT
/*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
  2 JOB-TITLE
  2 ADDRESS-LINE (2)
END-DEFINE
LIMIT 1
READ EMPLOY-VIEW BY CITY
/*****
WRITE NOTITLE 'EXAMPLE 1:'
      // 'RESULT OF WRITE STATEMENT:'
WRITE      / NAME ', ' FIRST-NAME ':' JOB-TITLE '*' (30)
WRITE      / 'RESULT OF PRINT STATEMENT:'
PRINT     / NAME ', ' FIRST-NAME ':' JOB-TITLE '*' (30)
/*****
WRITE      // 'EXAMPLE 2:'
      // 'RESULT OF WRITE STATEMENT:'
WRITE      / NAME 60X ADDRESS-LINE (1:2)
WRITE      / 'RESULT OF PRINT STATEMENT:'
PRINT     / NAME 60X ADDRESS-LINE (1:2)
/*****
END-READ
END

```

```

EXAMPLE 1:

RESULT OF WRITE STATEMENT:

SENKO           , WILLIE           : PROGRAMMER
*****

RESULT OF PRINT STATEMENT:

SENKO , WILLIE : PROGRAMMER *****

EXAMPLE 2:

RESULT OF WRITE STATEMENT:

SENKO
2200 COLUMBIA PIKE      #914

RESULT OF PRINT STATEMENT:

SENKO
PIKE #914                                     2200 COLUMBIA

```